

Pensieve header: Implementing the Jones polynomial.

Based on <http://drorbn.net/syd3>

```

In[ ]:= K = PD[X[1, 5, 2, 4], X[5, 3, 6, 2], X[3, 1, 4, 6]]

In[ ]:= t1 = K /. X[i_, j_, k_, l_] -> A * p[i, j] * p[k, l] + B * p[i, l] p[j, k]

In[ ]:= t2 = Expand[t1 /. PD -> Times]

In[ ]:= {7, 1} /. {{0, n_} -> n, {k_, n_} -> {k - 1, n * k}}

In[ ]:= {7, 1} //. {{0, n_} -> n, {k_, n_} -> {k - 1, n * k}}

In[ ]:= t3 = t2 //. p[i_, j_] p[j_, k_] -> p[i, k]

In[ ]:= t4 = t3 /. {p[i_, i_] -> d, p[i_, j_] ^2 -> d}

In[ ]:= t5 = Expand[t4 /. {B -> 1 / A, d -> -A^2 - 1 / A^2}]

In[ ]:= Knots

In[ ]:= Knot[3, 1] /. Knots

In[ ]:= Knot[10, 165] /. Knots

In[ ]:= K = Knot[10, 73] /. Knots;
t1 = K /. X[i_, j_, k_, l_] -> A * p[i, j] * p[k, l] + B * p[i, l] p[j, k];
t2 = Expand[t1 /. PD -> Times];
t3 = t2 //. {p[i_, j_] p[j_, k_] -> p[i, k], p[i_, j_] p[k_, j_] -> p[i, k]};
t4 = t3 /. {p[i_, i_] -> d, p[i_, j_] ^2 -> d};
Expand[t4 /. {B -> 1 / A, d -> -A^2 - 1 / A^2}]

In[ ]:= KB[pd_PD] := Module[{p, t1, t2, t3, t4, B, d},
  SetAttributes[p, Orderless];
  t1 = pd /. X[i_, j_, k_, l_] -> A * p[i, j] * p[k, l] + B * p[i, l] p[j, k];
  t2 = Expand[t1 /. PD -> Times];
  t3 = t2 //. {p[i_, j_] p[j_, k_] -> p[i, k]};
  t4 = t3 /. {p[i_, i_] -> d, p[i_, j_] ^2 -> d};
  Expand[t4 /. {B -> 1 / A, d -> -A^2 - 1 / A^2}]
]

In[ ]:= KB[Knot[8, 17] /. Knots]

In[ ]:= tab10 = Table[
  Echo[Timing[{k, KB[Knot[10, k] /. Knots]}]],
  {k, 165}
]

In[ ]:= Plus @@ tab10

In[ ]:= KB[GST48 /. Knots]

```